The Penultimate Alexander Invariant

A Definition of the MVA (From [Ar])



$\rightarrow$

	a	b	c
c	-1	$1-x$	y



$\rightarrow$

	a	b	c
c	$-y$	$x-1$	1


Joint with Jana Archibald

$$A = \frac{(-)^{i+j}\det(M_i^j)}{w_i(t_i-1)}\prod_k t_k^{\frac{\mathrm{rot}(k)-\mu(k)}{2}}$$



$\rightarrow$

	a_1	a_2	a_3	a_4	a_5
a_1	$y-1$	0	0	0	$1-x$
a_2	$1-y$	x	0	0	-1
a_3	$-y$	0	1	$y-1$	0
a_4	0	0	$-y$	1	$y-1$
a_5	0	0	$y-1$	$-y$	1

$\rightarrow \quad xy(1-y+y^2)$

Relations by J. Murakami (From [MJ])






J. Murakami



The Naik-Stanford Double Delta Relation (From [NS])



image not found

S. Naik T. Stanford




H. Murakami

(From [MH])

A Relation by H. Murakami



A Relation by A. Vaintrob (From [Va])

There's Lots More!

"God created the knots, all else in topology is the work of mortals"

Leopold Kronecker (paraphrased)

Visit!

Edit!

http://katlas.org

This handout and further links are at http://www.math.toronto.edu/~drorbn/Talks/Sandbjerg-0810/

Our Goal. Prove all these relations uniformly, at maximal confidence and minimal brain utilization.

$\Rightarrow$ We need an "Alexander Invariant" for arbitrary tangles, easy to define and compute and well-behaved under tangle compositions; better, "virtual tangles".

Circuit Algebras



A J-K Flip Flop

* Have "circuits" with "ends",
* Can be wired arbitrarily.
* May have "relations" - de-Morgan, etc.

Example $V\mathcal{T} = CA\langle\times,\times\rangle/R23 = PA\langle\times,\times,\times\rangle/R23, VR123, MR3$

Reminders from linear algebra. If X is a (finite) set,

$$\Lambda^k(X) := \langle k\text{-tuples in } X, \text{ modulo anti-symmetry}\rangle$$

$$\Lambda^{\mathrm{top}}(X) := \langle |X|\text{-tuples in } X, \text{ modulo anti-symmetry}\rangle$$

$$\Lambda^{1/2}(X) := \langle (|X|/2)\text{-tuples in } X, \text{ modulo anti-symmetry}\rangle.$$

If $Y \subset X^m$, the "interior multiplication" $i_Y : \Lambda^k(X) \to \Lambda^{k-m}(X)$ is anti-symmetric in Y.

Definition. An "Alexander half density with input strands X^{in} and output strands X^{out}" is an element of

$$\mathrm{AHD}(X^{\mathrm{in}}, X^{\mathrm{out}}) := \Lambda^{\mathrm{top}}(X^{\mathrm{out}}) \otimes \Lambda^{1/2}(X^{\mathrm{in}} \cup X^{\mathrm{out}}).$$

Often we extend the coefficients to some polynomial ring without warning.

Definition. If $\alpha_i \otimes p_i \in \mathrm{AHD}(X_i^{\mathrm{in}}, X_i^{\mathrm{out}})$ (for $i = 1, 2$), and $G = (X_1^{\mathrm{in}} \cup X_2^{\mathrm{in}}) \cap (X_1^{\mathrm{out}} \cup X_2^{\mathrm{out}})$ is the set of "gluable legs", the "gluing" in $\mathrm{AHD}(X_1^{\mathrm{in}} \cup X_2^{\mathrm{in}} - G, X_1^{\mathrm{out}} \cup X_2^{\mathrm{out}} - G)$ is

$$i_G(\alpha_1 \wedge \alpha_2) \otimes i_G(p_1 \wedge p_2).$$

Claim. This makes AHD a circuit algebra.

Definition. The "Penultimate Alexander Invariant" is defined using

$$pA : \overset{k\ \ j}{\underset{l\ \ i}{\times}} \mapsto (j \wedge k) \otimes \begin{pmatrix} l \wedge i + (t_i - 1) l \wedge j - t_l l \wedge k \\ +i \wedge j + t_l j \wedge k \end{pmatrix}$$

$$pA : \overset{l\ \ k}{\underset{i\ \ j}{\times}} \mapsto (k \wedge l) \otimes \begin{pmatrix} t_j i \wedge j - t_j i \wedge l + j \wedge k \\ +(t_i - 1) j \wedge l + k \wedge l \end{pmatrix}$$

Why Works?



Every "rook arrangement" in the above picture must have exactly l rooks in the yellow zone and l rooks in the purple zone. So for T_1 we only care about the minors in which exactly l of the $2l$ middle columns are dropped, and the rest is signs...

Weaknesses. Exponential, no understanding of cablings, no obvious "meaning". The ultimate Alexander invariant should address all that...

Challenge. Can you categorify this?

We Mean Business

```
1         (* WP: Wedge Product *)
2 WSort[expr_] := Expand[expr /. w_W :> Signature[w]*Sort[w]];
3 WP[0, _] = WP[_, 0] = 0;
4 WP[a_, b_] := WSort[Distribute[a ** b] /.
5     (c1_. * w1_W) ** (c2_. * w2_W) :> c1 c2 Join[w1, w2]];
6
7         (* IM: Interior Multiplication *)
8 IM[{}, expr_] := expr;
9 IM[i_, w_W] := If[FreeQ[w, i], 0,
10     -(-1)^Position[w, i][[1,1]]*DeleteCases[w, i] ];
11 IM[{is___, i_}, w_W] := IM[{is}, IM[i, w]];
12 IM[is_List, expr_] := expr /. w_W :> IM[is, w]
13
14        (* pA on Crossings *)
15 pA[Xp[i_,j_,k_,l_]] := AHD[(t[i]==t[k])(t[j]==t[l]), {i,l}, W[j,k],
16     W[l,i] + (t[i]-1)W[l,j] - t[l]W[l,k] + W[i,j] + t[l]W[j,k] ];
17 pA[Xm[i_,j_,k_,l_]] := AHD[(t[i]==t[k])(t[j]==t[l]), {i,j}, W[k,l],
18     t[j]W[i,j] - t[j]W[i,l] + W[j,k] + (t[i]-1)W[j,l] + W[k,l] ]
19
20        (* Variable Equivalences *)
21 ReductionRules[Times[]] = {};
22 ReductionRules[Equal[a_, b__]] := (# -> a)& /@ {b};
23 ReductionRules[eqs_Times] := Join @@ (ReductionRules /@ List@@eqs)
24
25        (* AHD: Alexander Half Densities *)
26 AHD[eqs_, is_, -os_, p_] := AHD[eqs, is, os, Expand[-p]];
27 AHD /: Reduce[AHD[eqs_, is_, os_, p_]] :=
28   AHD[eqs, Sort[is], WSort[os], WSort[p /. ReductionRules[eqs]]];
29 AHD /: AHD[eqs1_,is1_,os1_,p1_] AHD[eqs2_,is2_,os2_,p2_] := Module[
30   {glued = Intersection[Union[is1, is2], List@@Union[os1, os2]]},
31   Reduce[AHD[
32     eqs1*eqs2 //. eq1_Equal*eq2_Equal /;
33       Intersection[List@@eq1, List@@eq2] =!= {} :> Union[eq1, eq2],
34     Complement[Union[is1, is2], glued],
35     IM[glued, WP[os1, os2]],
36     IM[glued, WP[p1, p2]]
37 ]] ]
38
39        (* pA on Circuit Diagrams *)
40 pA[cd_CircuitDiagram, eqs___] := pA[cd, {}, AHD[Times[eqs], {}, W[], W[]]];
41 pA[cd_CircuitDiagram, done_, ahd_AHD] := Module[
42   {pos = First[Ordering[Length[Complement[List @@ #, done]] & /@ cd]]},
43   pA[Delete[cd, pos], Union[done, List @@ cd[[pos]]], ahd*pA[cd[[pos]]]]
44 ];
45 pA[CircuitDiagram[], _, ahd_AHD] := ahd
```

Comments online **2.** `W[i1,i2,...]` represents $i_1 \wedge i_2 \wedge \ldots$. To sort it we `Sort` its arguments and multiply by the `Signature` of the permutation used. **3.** The wedge product of 0 with anything is 0. **4-5.** The wedge product of two things involves applying the `Distribute`ive law, `Join`ing all pairs of `W`'s, and `WSort`ing the result. **8.** Inner multiplying by an empty list of indices does nothing. **9-10.** Inner multiplying a single index yields 0 if that index is not pressent, otherwise it's a sign and the index is deleted. **11-12.** Aftwrwards it's simple recursion. **15-18.** For the crossings `Xp` and `Xm` it is straightforward to determine the incoming strands, the outgoing ones, and the variable equivalences. The associated half-densities are just as in the formulas. **21-23.** The technicalities of imposing variable equivalences are annoying. **26.** That's all we need from the definition of a tensor product. **27-28.** Straightforward simplifications. **29.** The (circuit algebra) product of two Alexander Half Densities: **30.** The `glued` strands are the intersection of the ins and the outs. **32-33.** Merging the variable equivalences is tricky but natural. **34-35.** Removing the `glued` strands from the ins and outs. **36 The Key Point.** The wedge product of the half-densities, inner with the `glued` strands. **40-45.** A quick implementation of a "thin scanning" algorithm for multiple products. The key line is **42**, where we select the next crossing we multiply in to be the crossing with the fewest "loose strands".

▪ Overcrossings Commute



Hence "w-knots"

```
In[15]:= Equal[
           pA[CircuitDiagram[Xp[1, 7, 4, 6], Xp[2, 3, 5, 7]]],
           pA[CircuitDiagram[Xp[2, 7, 5, 6], Xp[1, 3, 4, 7]]] ]
Out[15]= True
```



```
In[10]:= Timing[res4 = pA[#, (t[1] == t[6]) (t[11] == t[16]) (t[21] == t[26])] & /@ {
           CircuitDiagram[
            Xm[19, 1, 20, 2], Xp[11, 3, 12, 2], Xp[3, 30, 4, 29], Xm[4, 21, 5, 22],
            Xp[6, 23, 7, 22], Xm[7, 28, 8, 29], Xm[12, 8, 13, 9], Xp[18, 10, 19, 9],
            Xm[27, 13, 28, 14], Xp[23, 15, 24, 14], Xm[24, 16, 25, 17], Xp[26, 18, 27, 17]
           ],
           CircuitDiagram[
            Xp[1, 28, 2, 27], Xm[2, 23, 3, 24], Xm[17, 3, 18, 4], Xp[13, 5, 14, 4],
            Xm[14, 6, 15, 7], Xp[16, 8, 17, 7], Xp[8, 25, 9, 24], Xm[9, 26, 10, 27],
            Xm[29, 11, 30, 12], Xp[21, 13, 22, 12], Xm[22, 18, 23, 19], Xp[28, 20, 29, 19]
           ]
          }]

         A very large output was generated. Here is a sample of it:
         {9.86, {AHD[(t[1] == t[2] == t[3] == t[4] == t[5] == t[6] == t[7] == t[8] == t[9] == t[10])
             (t[11] == t[12] == t[13] == t[14] == t[15] == t[16] == t[17] == t[18] == t[19] == t[20])
             (t[21] == t[22] == t[23] == t[24] == t[25] == t[26] == t[27] == t[28] == t[29] == t[30]),
Out[10]=   {1, 6, 11, 16, 21, 26}, <<1>>,
           -t[1]^2 t[11]^2 t[21]^2 W[1, 5, 6, 11, 15, 21] + <<2574>>], <<1>>}}
         Show Less   Show More   Show Full Output   Set Size Limit...

In[11]:= Equal @@ (Last /@ res4)
Out[11]= True
```

(The program also prints "False" when appropriate, and computes Alexander polynomials)
More at http://www.math.toronto.edu/~drorbn/Sandbjerg-0810/pA.nb

▪ Commutators Commute



```
In[5]:= Equal[
          pA[CircuitDiagram[Xp[1, 2, 11, 8], Xm[11, 3, 12, 7],
            Xp[12, 4, 13, 10], Xm[13, 5, 6, 9]], t[2] == t[3], t[4] == t[5]],
          pA[CircuitDiagram[Xp[1, 4, 11, 10], Xm[11, 5, 12, 9],
            Xp[12, 2, 13, 8], Xm[13, 3, 6, 7]], t[2] == t[3], t[4] == t[5]] ]
Out[5]= True
```

Question.

Does this specify the Alexander polynomial?

References

[Ar] J. Archibald, *The Weight System of the Multivariable Alexander Polynomial,* arXiv:0710.4885.

[MH] H. Murakami, *A Weight System Derived from the Multivariable Conway Potential Function,* Jour. of the London Math. Soc. **59** (1999) 698–714, arXiv:math/9903108.

[MJ] J. Murakami, *A State Model for the Multi-Variable Alexander Polynoomial,* Pac. Jour. of Math. **157-1** (1993) 109–135.

[NS] S. Naik and T. Stanford, *A Move on Diagrams that Generates S-Equivalence of Knots,* Jour. of Knot Theory and its Ramifications **12-5** (2003) 717–724, arXiv:math/9911005.

[Va] A. Vaintrob, *Melvin-Morton Conjecture and Primitive Feynman Diagrams,* Inter. J. Math. **8** (1997) 537–553, arXiv:q-alg/9605028.